Formalizing David Hilbert's Foundations of Geometry

Jürgen Uhl

September 10, 2023

Contents

theory *HilbertGeometry*
imports *Main*
begin

1 Introduction

This theory formalizes David Hilbert's Geometry Axioms and Theorems using the Proof System Isabelle. Chapters correspond to the axiom groups

- Group I: Axioms of Connection
- Group II: Axioms of Order
- Group III: Axioms of Parallels (Euclids axiom)
- Group IV: Axioms of Congruence
- Group V: Axioms of Continuity (Archimedess axiom)

Each chapter starts out with the formalization of the axioms followed by the proofs of the given theorems.

end
theory *Connection*
imports *Main*
begin

2 Group I: Axioms of connection

2.1 Axioms

Axioms of connection formalize the relationship between *point*s, (*straight*) *line*s and *plane*s. We model points by using a type variable *'point* and lines and planes as sets of points. The axioms then characterize the predicates *Line* and *Plane*.

Shorthand for two distinct points lying in the same line or plane

abbreviation $in2\ a\ A\ B \equiv A \neq B \wedge A \in a \wedge B \in a$

Shorthand for three (not necessarily distinct) points lying in the same line or plane

abbreviation $in3\ a\ A\ B\ C \equiv A \in a \wedge B \in a \wedge C \in a$

abbreviation $in4\ a\ A\ B\ C\ D \equiv A \in a \wedge B \in a \wedge C \in a \wedge D \in a$

```
locale Connection =
  fixes Line :: 'point set ⇒ bool
    and Plane :: 'point set ⇒ bool
  assumes I-1-2: A ≠ B ⟹ ∃!a. Line a ∧ A ∈ a ∧ B ∈ a
— I.1 For any two points there exists a straight line passing through them.
— I.2 There exists only one straight line passing through any two distinct points.
    and I-3a: Line a ⟹ ∃ A B. in2 a A B
— I.3a At least two points lie on any straight line. ...
    and I-3b: ∃ A B C. ∄ a. Line a ∧ in3 a A B C
— I.3b ... There exist at least three points not lying on the same straight line.
    and I-4a-5: ∄ a. Line a ∧ in3 a A B C ⟹ ∃!α. Plane α ∧ in3 α A B C
— I.4a There exists a plane passing through any three points not lying on the same straight line. ...
— I.5 There exists only one plane passing through any three points not lying on the same straight line.
    and I-4b: Plane α ⟹ α ≠ {}
— I.4b ... At least one point lies on any given plane.
    and I-6: ⟦Line a; Plane α; in2 a A B; in2 α A B⟧ ⟹ a ⊆ α
— I.6 If two points A and B of a straight line a lie in a plane α, then all points of a lie in α.
    and I-7: ⟦Plane α; Plane β; A ∈ α; A ∈ β⟧ ⟹
              ∃ B. B ≠ A ∧ B ∈ α ∧ B ∈ β
— I.7 If two planes have one point in common, then they have at least one more point in common.
    and I-8: ∃ A B C D. ∄ α. Plane α ∧ A ∈ α ∧ B ∈ α ∧ C ∈ α ∧ D ∈ α
— I.8 There exist at least four points not lying in the same plane.

context Connection
begin

— Note that in-line A B C does not require A B C to be distinct
abbreviation in-line A B C ≡ ∃ a. Line a ∧ in3 a A B C

abbreviation in-line4 A B C D ≡ ∃ a. Line a ∧ in4 a A B C D

abbreviation distinct3 A B C ≡ A≠B ∧ A≠C ∧ B≠C

abbreviation distinct4 A B C D ≡ A≠B ∧ A≠C ∧ A≠D ∧ B≠C ∧ B≠D ∧ C≠D

lemma lemma-I-1: card a ≥ 2 ⟹ ∃ A B. in2 a A B
  by (metis Suc-diff-le Suc-eq-plus1 Zero-not-Suc
      cancel-comm-monoid-add-class.diff-cancel card.empty is-singletonI'
      is-singleton-altdef not-numeral-le-zero one-add-one)

lemma lemma-I-2: ⟦Line a; in2 a A B; C ∉ a⟧ ⟹ ¬in-line A B C
  using I-1-2 by auto

lemma lemma-I-3: ⟦Plane α; Plane β; Line a; in2 a A B; C ∉ a;
                  in3 α A B C; in3 β A B C⟧ ⟹ α = β
  using lemma-I-2 I-4a-5 by blast

lemma lemma-I-4: ⟦Plane α; A ≠ B; ¬in-line A B C; in3 α A B C;
                  in-line A B D⟧ ⟹ D ∈ α using I-6 by fastforce

lemma lemma-I-5: ¬in-line A B C ⟹ distinct3 A B C
  using I-1-2 I-4a-5 I-8 by (metis (full-types))

theorem theorem-I-1a:
  fixes a b
  assumes Line a Line b a ≠ b
  shows card (a ∩ b) < 2
— Two straight lines of a plane have either one point or no point in common; ... We have generalized the theorem to two arbitrary lines, whether they lie in the same plane or not.
proof (rule ccontr)
  assume ¬card (a ∩ b) < 2
  hence card (a ∩ b) ≥ 2 by auto
  hence ∃ A B. in2 (a ∩ b) A B using lemma-I-1 by blast
```

```
  hence a = b using I-1-2 ‹Line a› ‹Line b› by blast
  thus False  using ‹a ≠ b› by simp
qed

theorem theorem-I-1b:
  fixes α β
  assumes Plane α Plane β α ≠ β
  shows (α ∩ β = {}) ∨ (∃ a. Line a ∧ α ∩ β = a)
— ... two planes have no point in common or a straight line in common; ...
proof (rule ccontr)
  assume a1: ¬(α ∩ β = {} ∨ (∃ a. Line a ∧ α ∩ β = a))
  hence α ∩ β ≠ {} ∧ (∄ a. Line a ∧ α ∩ β = a) by simp
  thus False
  proof
    assume α ∩ β ≠ {}
    then obtain A where oA: A ∈ α ∧ A ∈ β by blast
    then obtain B where oB: A ≠ B ∧ B ∈ α ∧ B ∈ β
      using I-7 by (metis assms(1−2))
    then obtain a where ex: Line a ∧ A ∈ a ∧ B ∈ a using I-1-2 by blast
    hence l-ss-a12: a ⊆ α ∩ β using I-6 assms(1−2) oA oB by blast
    from ex have α ∩ β ⊆ a using lemma-I-3 assms oA oB by blast
    from this l-ss-a12 have α ∩ β = a by auto
    thus False using a1 ex by auto
  qed
qed

theorem theorem-I-1c:
  fixes α a
  assumes Line a Plane α ¬ a ⊆ α
  shows card (α ∩ a) < 2
— ... a plane and a straight line not lying in it have no point or one point in common.
proof (rule ccontr)
  assume ¬card (α ∩ a) < 2
  hence card (α ∩ a) ≥ 2 by auto
  hence ∃  A B. in2 (α ∩ a) A B using lemma-I-1 by blast
  hence a ⊆ α using I-6 assms(1) assms(2) by blast
  thus False using assms(3) by simp
qed

theorem theorem-I-2a:
  fixes a A
  assumes Line a A ∉ a
  shows ∃!α. Plane α ∧ a ⊆ α ∧ A ∈ α
— Through a straight line and a point not lying in it, ..., one and only one plane may be made to pass.
proof −
  obtain B C where oBC:  in2 a B C using I-3a assms(1) by blast
  have i1: ¬in-line A B C
    using oBC lemma-I-2 assms by (smt (verit, best) insert-commute)
  hence ∃!α. Plane α ∧ in3 α A B C using  I-4a-5 i1 by simp
  thus ?thesis using I-6 oBC assms(1)
    by (smt (verit) empty-iff in-mono insert-iff subsetI)
qed

theorem theorem-I-2b:
  fixes a b A
  assumes Line a Line b a ≠ b A ∈ a A ∈ b
  shows ∃!α. Plane α ∧ a ⊆ α ∧ b ⊆ α
— ..., or through two distinct straight lines having a common point, one and only one plane may be made to pass.
proof −
  obtain B where oB: in2 a A B using I-3a assms(1) assms(4) by metis
  hence i: B ∉ b using I-1-2 assms by auto
  obtain C where oC: in2 b A C using I-3a assms(2) assms(5) by metis
  hence aC: C ∉ a using I-1-2 assms by auto
  hence ¬in-line A B C using  lemma-I-2 assms oB by blast
```

hence $\exists!\alpha$. *Plane* $\alpha \wedge$ *in3* α *A B C* **using** *I-4a-5 i assms* **by** *simp*
thus *?thesis* **using** *I-6 oB oC assms(1−2) aC theorem-I-2a*
by (*smt* (*verit*, *ccfv-threshold*) *in-mono*)
qed

end
end
theory *Order*
imports *Connection*
begin

3 Group II: Axioms of order

3.1 Axioms

locale *Order* = *Connection Line* **for** *Line* :: *'point set* $\Rightarrow$ *bool* +
fixes *btw* :: *'point* $\Rightarrow$ *'point* $\Rightarrow$ *'point* $\Rightarrow$ *bool*
assumes *II-1a*: *btw A B C* $\Longrightarrow$ *distinct3 A B C* $\wedge$ *in-line A B C*
— If a point B lies between a point A and a point C, then A, B, and C are distinct points on the same straight line ...
and *II-1b*: *btw A B C* $\Longrightarrow$ *btw C B A*
— ... and B also lies between C and A.
and *II-2*: $A \neq B \Longrightarrow \exists C$. *btw A B C*
— For any two points A and B on the straight line AB, there exists at least one point C such that the point B lies between A and C.
and *II-3*: $[\![$*Line a*; *in3 a A B C*$]\!] \Longrightarrow \neg$(*btw A B C* $\wedge$ *btw B A C*) $\wedge$
$\neg$(*btw A B C* $\wedge$ *btw A C B*) $\wedge$ $\neg$(*btw B A C* $\wedge$ *btw B C A*)
— Out of any three points on the same straight line there exists not more than one point lying between the other two.
and *II-4*: $[\![$*Plane* α; $\neg$*in-line A B C*; *in3* α *A B C*;
Line a; $a \subseteq \alpha \wedge A \notin a \wedge B \notin a \wedge C \notin a$;
btw A D B $\wedge D \in a]\!] \Longrightarrow (\exists E.\ E \in a \wedge$ (*btw A E C* $\vee$ *btw B E C*))
— Let A, B, and C be three points not lying on the same straight line, and let a be a straight line in the plane ABC not passing through any of the points A, B, or C. Then, if the straight line a passes through an interior point of the segment AB, it also passes through an interior point of the segment AC or through an interior point of the segment BC.

context *Order*
begin

lemma *lemma-II-1*: $[\![\neg$*in-line A B C*; *btw A D B*$]\!] \Longrightarrow C \neq D$ **using** *II-1a* **by** *blast*

lemma *lemma-II-2*: $[\![$*distinct3 A B C*; $\neg$*in-line A B C*; *btw A D B*$]\!]$
$\Longrightarrow \neg$*in-line A D C*
using *II-1a lemma-I-2* **by** *blast*

The following scenarios (lemma II-3a and b) are used multiple times in the subsequent theorems: a) Have a line A B C such that A is not between B and C. Have D not on AC and E such that D is between C and E. Then the line AD intersects EB in a point F which is between E and B. Note that CDE and BFE can be interchanged.



lemma *lemma-II-3a*:
fixes *A B C D E*
assumes *distinct3 A B C in-line A B C* $\neg$*btw B A C*
$\neg$*in-line A D B distinct3 C D E btw C D E*
shows $\exists F$. *in-line A F D* $\wedge$ *btw E F B*
proof −

```
  have BEC: ¬in-line B E C
    using assms(1,2,4−6) II-1a lemma-I-2 lemma-I-5 by meson
  obtain α where oα: Plane α ∧ in3 α A D B using I-4a-5 assms(4) by meson
  obtain a where oa: Line a ∧ A ∈ a ∧ D ∈ a
    using II-1a I-1-2 assms(6) by metis
  then have aα: a ⊆ α using oα assms(2,4) I-6 by metis
  have Ca: C ∉ a using assms(1,2,4) oa lemma-I-2 by blast
  then have Ea: E ∉ a  using  assms(6) oa II-1a II-1b lemma-I-2 by meson
  have Ba: B ∉ a using assms(1,2,4) oa lemma-I-2 by blast
  have becα: in3 α C E B using oα II-1a I-6 assms(1,2,5,6) by blast
  obtain F where oF: F ∈ a ∧ (btw B F C ∨ btw E F B)
    using BEC aα oα oa Ba Ea Ca becα assms(6) II-1b II-4[of α C E B a D]
    by blast
  then have ¬in-line B F C using BEC Ba II-1a assms(2,3) lemma-I-2 oa
    by (smt(verit, ccfv-threshold))
  then have ex: ∃F. in-line A D F ∧ btw E F B using II-1a oF oa by blast
  then show ?thesis by blast
qed

abbreviation lemma-II-3a-assms A B C D E ≡ distinct3 A B C ∧ in-line A B C ∧
¬btw B A C ∧ ¬in-line A D B ∧ distinct3 C D E ∧ btw C D E
```

b) Have two lines A B C and C D E such that A is not between B and C and E is not between C and D. Then the line AD intersects BE in a point F which is between A and D and between E and B.




```
lemma lemma-II-3b:
  fixes A B C D E
  assumes btw A B C ¬in-line A D C btw C D E
  shows∃F. btw A F D ∧ btw E F B
proof −
  have assmsABCDE: lemma-II-3a-assms A B C D E
    using assms II-1a II-3 lemma-I-2 by (smt(verit))
  then obtain F where oF: in-line A F D ∧ btw E F B
    using lemma-II-3a[of A B C D E] by auto
  have assmsEDCBA: lemma-II-3a-assms E D C B A
    using assms II-1a II-1b II-3 lemma-I-2 by (smt(verit))
  then obtain G where oG: in-line E G B ∧ btw A G D
    using lemma-II-3a[of E D C B A] by auto
  have F = G using assms(1,2) oF oG II-1a lemma-I-2 by (smt(verit, best))
  then show ?thesis using oF oG by auto
qed

abbreviation lemma-II-3b-assms A B C D E ≡
btw A B C ∧ ¬in-line A D C ∧ btw C D E
```

Choose E not on the line AC, have F such that btw A E F, then G such that btw F C G and then apply lemma II-3 to obtain a D where btw A D C.



theorem *theorem-II-1*:
  fixes *A C*
  assumes *A* ≠ *C*
  shows ∃ *D*. *btw A D C*
— Between any two points A and C of a straight line, there always exists at least one point D on the line AC which is between A and C.
proof –
  obtain *E* **where** *oE*: ¬*in-line A E C* **using** *assms I-1-2 I-3b* **by** *metis*
  then obtain *F* **where** *oF*: *btw A E F* **using** *assms I-1-2 II-2* **by** *metis*
  then obtain *G* **where** *oG*: *btw F C G*
    using *oE I-1-2 II-1a II-2* **by** (*metis* (*full-types*))
  then obtain *D* **where** *btw A D C*
    using *oE oF oG II-1a II-1b II-3 lemma-I-2 lemma-II-3a*[*of G C F E A*]
    by (*smt*(*verit*))
  then show *?thesis* **by** *auto*
qed

Assuming that neither btw A B C nor btw A C B we prove that btw B A C. Choose D not on the line BC, have G such that btw A D G, then show that btw C E G and btw B F G, then btw B D E and finally btw B A C.



theorem *theorem-II-2*:
  fixes *A B C*
  assumes *distinct3 A B C in-line A B C*
  shows *btw B A C* ∨ *btw A B C* ∨ *btw A C B*
— Among three points A, B and C on a straight line there is one lying between the two others.
proof (*rule disjCI*)
  assume ¬(*btw A B C* ∨ *btw A C B*)
  then have *i*: ¬*btw A B C* ∧ ¬*btw B C A* **using** *II-1b* **by** *blast*
  then obtain *D* **where** *oD*: ¬*in-line B D C* **using** *assms*(*1*) *I-1-2 I-3b* **by** *metis*
  then obtain *G* **where** *oG*: *btw A D G* **using** *assms I-1-2 II-2* **by** *metis*
  obtain *E* **where** *oE*: *in-line B D E* ∧ *btw C E G*
    using *assms i oD oG II-1a II-1b lemma-II-3a*[*of B C A D G*] **by** *blast*
  obtain *F* **where** *oF*: *in-line C D F* ∧ *btw B F G*
    using *assms i oD oG II-1a II-1b lemma-II-3a*[*of C B A D G*] **by** *blast*
  then have *bBDE*: *btw B D E*
    using *oD oG oE oF II-1a II-1b II-3 lemma-I-2 lemma-II-3a*[*of C E G F B*]
    by (*smt*(*verit*))
  have *in-line G D A* ∧ *in-line B A C* **using** *assms*(*2*) *lemma-II-1 oG* **by** *blast*
  then show *btw B A C* **using** *assms*(*2*) *oD oE II-1a bBDE II-1b II-3*
      lemma-I-2 lemma-II-3a[*of G C E D B*] **by** (*smt*(*verit*))
qed

lemma *lemma-II-4*:
  fixes *A B C*

assumes *distinct3 A B C in-line A B C*
shows $\exists D\ E\ F.\ \{D, E, F\} = \{A, B, C\} \wedge$ *btw D E F*
proof –
consider (*BAC*) *btw B A C* | (*ABC*) *btw A B C* | (*ACB*) *btw A C B*
using *assms theorem-II-2* **by** *auto*
then show *?thesis*
proof *cases*
case *BAC*
obtain *D E F* **where** $D{=}B \wedge E{=}A \wedge F{=}C$ **by** *simp*
then have $\{D, E, F\} = \{A, B, C\} \wedge$ *btw D E F* **using** *BAC* **by** *auto*
then show *?thesis* **by** *auto*
next
case *ABC*
obtain *D E F* **where** $D{=}A \wedge E{=}B \wedge F{=}C$ **by** *simp*
then have $\{D, E, F\} = \{A, B, C\} \wedge$ *btw D E F* **using** *ABC* **by** *auto*
then show *?thesis* **by** *auto*
next
case *ACB*
obtain *D E F* **where** $D{=}A \wedge E{=}C \wedge F{=}B$ **by** *simp*
then have $\{D, E, F\} = \{A, B, C\} \wedge$ *btw D E F* **using** *ACB* **by** *auto*
then show *?thesis* **by** *auto*
qed
qed

Find E not on AD and F with btw C E F, then apply lemma II-3b to find G with btw A G E and btw F G B. With lemma II-3a, first have H with btw G H D and, finally, find I=C with btw A I D.



lemma *lemma-II-5a*:
fixes *A B C D*
assumes *btw A B C btw B C D*
shows *btw A C D*
— If btw A B C and btw B C D then btw A C D. By symmetry, lemma II-5 below also shows btw A B D.
proof –
have *ABCD*: *in-line4 A B C D*
using *assms II-1a lemma-I-2* **by** (*smt*(*verit, best*))
obtain *E* **where** *oE*: ¬*in-line A E B*
using *assms I-1-2 I-3b II-1a* **by** *metis*
then obtain *F* **where** *oF*: *btw C E F* **using** *assms ABCD I-1-2 II-2* **by** *metis*
have *lemma-II-3b-assms A B C E F*
using *assms*(*1*) *oE oF II-1a lemma-I-2* **by** *meson*
then obtain *G* **where** *oG*: *btw A G E* ∧ *btw F G B*
using *lemma-II-3b*[*of A B C E F*] **by** *auto*
have *assmsFBGCD*: *lemma-II-3a-assms F G B C D*
using *assms*(*2*) *oE oF oG II-1a II-1b II-3 lemma-I-2* **by** (*smt*(*verit*))
then obtain *H* **where** *oH*:*in-line F C H* ∧ *btw D H G*
using *lemma-II-3a*[*of F G B C D*] **by** *auto*
have *assmsEAGHD*: *lemma-II-3a-assms E A G H D*
using *ABCD oE oF oG oH II-1a II-1b II-3 lemma-I-2* **by** (*smt*(*verit, best*))
then obtain *J* **where** *oJ*: *in-line E H J* ∧ *btw D J A*
using *lemma-II-3a*[*of E A G H D*] **by** *auto*
then have *J* = *C*
using *assms*(*1*) *oF oH II-1a assmsFBGCD assmsEAGHD lemma-I-2* **by** (*smt*(*verit*))
then show *?thesis* **using** *oJ* **using** *II-1b* **by** *blast*
qed

lemma *lemma-II-5*:
  fixes *A B C D*
  assumes *btw A B C btw B C D*
  shows *btw A B D* ∧ *btw A C D*
  using *assms II-1b lemma-II-5a* **by** *blast*

As for lemma II-5, find E not on AD and F with btw C E F, and G with btw A G E and btw F G B. With lemma II-3a, first have H with btw G H D and, finally, find I=C with btw B I D. Then use lemma II-5 to show the second part of the conjunction.



lemma *lemma-II-6*:
  fixes *A B C D*
  assumes *btw A B C btw A C D*
  shows *btw B C D* ∧ *btw A B D*
— If btw A B C and btw A C D then btw B C D.
proof −
  have *ABCD*: *in-line4 A B C D*
    using *assms II-1a lemma-I-2* **by** (*smt*(*verit*, *best*))
  obtain *E* **where** *oE*: ¬*in-line A E B*
    using *assms I-1-2 I-3b II-1a* **by** *metis*
  then obtain *F* **where** *oF*: *btw C E F* **using** *ABCD II-2* **by** *metis*
  have *lemma-II-3b-assms A B C E F*
    using *assms*(*1*) *oE oF II-1a lemma-I-2* **by** *meson*
  then obtain *G* **where** *oG*: *btw A G E* ∧ *btw F G B*
    using *lemma-II-3b*[*of A B C E F*] **by** *auto*
  have *assmsEAGCD*: *lemma-II-3a-assms E G A C D*
    using *assms oE oF oG II-1a II-1b II-3 lemma-I-2* **by** (*smt* (*verit*))
  then obtain *H* **where** *oH*: *in-line E C H* ∧ *btw D H G*
    using *lemma-II-3a*[*of E G A C D*] **by** *auto*
  have *assmsFBGHD*: *lemma-II-3a-assms F B G H D*
    using *assms oE oF oG oH II-1a II-1b II-3 lemma-I-2* **by** (*smt*(*verit*))
  then obtain *I* **where** *oI*: *in-line F H I* ∧ *btw D I B*
    using *lemma-II-3a*[*of F B G H D*] **by** *auto*
  then have *I* = *C*
    using *assms oF oH assmsFBGHD II-1a lemma-I-2* **by** (*smt*(*verit*))
  then show *?thesis* **using** *assms oI II-1b lemma-II-5* **by** *blast*
qed

abbreviation *btw4 A B C D* ≡ *btw A B C* ∧ *btw A B D* ∧ *btw A C D* ∧ *btw B C D*

theorem *theorem-II-3*:
  fixes *A B C D*
  assumes *distinct4 A B C D in-line4 A B C D*
  shows ∃ *E F G H*. {*E*,*F*,*G*,*H*} = {*A*,*B*,*C*,*D*} ∧ *btw4 E F G H*
— For four points A B C D on a straight line can be arranged such that btw4 holds for this arrangement
proof −
  obtain *K L M* **where** *KLM*: {*K*,*L*,*M*} = {*A*,*B*,*C*} ∧ *btw K L M*
    using *assms lemma-II-4* **by** *metis*
  have *K*≠*L* **using** *assms*(*1*) *KLM* **by** (*metis insert-iff singletonD*)
  then have *distinct3 K L D* ∧ *in-line K L D* **using** *assms KLM* **by** *auto*
  then consider (*DKL*) *btw D K L* | (*KDL*) *btw K D L* | (*KLD*) *btw K L D*
    using *assms theorem-II-2* **by** *auto*
  then show *?thesis*

```
  proof cases
    case DKL
    then have {D,K,L,M} = {A,B,C,D} ∧ btw4 D K L M
      using KLM lemma-II-5[of D K L M] lemma-II-6[of D K L M] by auto
    then show ?thesis by auto
  next
    case KDL
    then have {K,D,L,M} = {A,B,C,D} ∧ btw4 K D L M
      using KLM lemma-II-5[of K D L M] lemma-II-6[of K D L M] by auto
    then show ?thesis by auto
  next
    case KLD
    have K≠M using assms(1) KLM by (metis insert-iff singletonD)
    then have distinct3 K M D ∧ in-line K M D using assms KLM by auto
    then consider (DKM) btw D K M | (KDM) btw K D M | (KMD) btw K M D
      using assms KLM theorem-II-2 by auto
    then show ?thesis
    proof cases
      case DKM
      then have False using KLD KLM II-1a II-1b lemma-II-6 by metis
      then show ?thesis by auto
    next
      case KDM
      then have {K,L,D,M} = {A,B,C,D} ∧ btw4 K L D M
        using KLM KLD lemma-II-5[of K L D M] lemma-II-6[of K L D M] by auto
       then show ?thesis by auto
    next
      case KMD
      then have {K,L,M,D} = {A,B,C,D} ∧ btw4 K L M D
        using KLM KLD lemma-II-5[of K L M D] lemma-II-6[of K L M D] by auto
       then show ?thesis by auto
    qed
  qed
qed

end
end
```